Example of Optimal Control

```
> restart:
> with(plots):
```

The target

```
> J := Int(-v(t),t=0..1) ;
```

$$J := \int_0^1 (-v(t))\, \mathrm{d}t \tag{1}$$

The ODE (or dynamical system)

```
> EQ1 := diff(x(t),t)=v(t) ;
  EQ2 := diff(v(t),t)=F(t) ;
```

$$EQ1 := \frac{\mathrm{d}}{\mathrm{d}t} x(t) = v(t)$$

$$EQ2 := \frac{\mathrm{d}}{\mathrm{d}t} v(t) = F(t) \tag{2}$$

The boundary conditions

```
> BC := x(0) = 0, v(0)=0, v(1)=0 ;
```

$$BC := x(0) = 0, v(0) = 0, v(1) = 0 \tag{3}$$

Limitation of the control

```
> Flimit := -1 <= F(t) and F(t) <= 1 ;
```

$$Flimit := -1 \leq F(t) \text{ and } F(t) \leq 1 \tag{4}$$

Approximate solution using Direct discretization

N is teh number of intervals, h = 1/N is the size of the intervals

Approximate the target with quadrature rule, dor example trapezoidal rule.

```
> JF := op(1,J) ;
```

$$JF := -v(t) \tag{1.1}$$

```
> Target := -h/2*sum( v[k+1]+v[k], k=0..N-1) ;
```

$$Target := -\frac{1}{2} h \left(v_N - v_0 + \sum_{k=0}^{N-1} 2 v_k \right) \tag{1.2}$$

Use finite difference (Explicit Euler) -> Finite Difference Becomes constraints of the problem

```
> FDE1 := (x[k+1]-x[k])/h - v[k] ;
  FDE2 := (v[k+1]-v[k])/h - F[k] ;
```

$$FDE1 := \frac{x_{k+1} - x_k}{h} - v_k$$

$$FDE2 := \frac{v_{k+1} - v_k}{h} - F_k \tag{1.3}$$

The (discrete) boundary conditions

```
> BCdiscrete := x[0] = 0, v[0]=0, v[N]=0 ;
```

(1.4)

$$BCdiscrete := x_0 = 0, v_0 = 0, v_N = 0 \quad (1.4)$$

We have now a NLP (non linear programming) problem:

minimize Target
subjet to the constraints FDE1 and FDE2 with BCdiscrete and Flimit

Solve NLP using maple Minimize function

```
> N := 50 ;
  h := 1/10 ;
```

$$N := 50$$

$$h := \frac{1}{10} \quad (1.5)$$

```
> SOL := Optimization[Minimize]( Target, {seq(FDE1=0,k=0..N-1),
                                          seq(FDE2=0,k=0..N-1),
                                          BCdiscrete,
                                          seq(F[k]<=1,k=0..N-1),
                                          seq(F[k]>=-1,k=0..N-1)} )
  :
> plot( subs(SOL[2],[seq([k,F[k]],k=0..N-1)]) ) ;
```



Use Pontryagin to compute the exact solution

First Step: Build the Hamiltonian

```
> H := rhs(EQ1)*lambda1(t) + rhs(EQ2)*lambda2(t) + op(1,J) ;
```

$$H := v(t)\,\lambda 1(t) + F(t)\,\lambda 2(t) - v(t) \quad (2.1)$$

```
> DIFF := proc( F, X )
    local tmp ;
    subs(tmp=X,diff(subs(X=tmp,F),tmp)) ;
  end;
```

$$DIFF := \mathbf{proc}(F, X)\ \mathbf{local}\ tmp;\ subs(tmp = X, diff(subs(X = tmp, F), tmp))\ \mathbf{end\ proc} \quad (2.2)$$

Build the BVP problem

```
> BVP1 := diff(x(t),t) = DIFF(H,lambda1(t)) ;
  BVP2 := diff(v(t),t) = DIFF(H,lambda2(t)) ;
  # the 2 co equation
  BVP3 := diff(lambda1(t),t) = -DIFF(H,x(t)) ;
  BVP4 := diff(lambda2(t),t) = -DIFF(H,v(t)) ;
```

$$BVP1 := \frac{\mathrm{d}}{\mathrm{d}t}\,x(t) = v(t)$$

$$BVP2 := \frac{\mathrm{d}}{\mathrm{d}t}\,v(t) = F(t)$$

$$BVP3 := \frac{\mathrm{d}}{\mathrm{d}t}\,\lambda 1(t) = 0$$

$$BVP4 := \frac{\mathrm{d}}{\mathrm{d}t}\,\lambda 2(t) = -\lambda 1(t) + 1 \quad (2.3)$$

The boundary condition

```
> fullBC := BC, lambda1(1)=0;
```

$$fullBC := x(0) = 0, v(0) = 0, v(1) = 0, \lambda 1(1) = 0 \quad (2.4)$$

The solution of argmin H is F(t) = +1 if lambda2(t) < 0, F(t)=-1 if lambda2(t) >0, if lambda2(t) = 0 F(t) is not determinate.

```
> SOLFeq1  := simplify(dsolve( subs(F(t)=1,{BVP||(1..4)}) )) ;
  SOLFeqm1 := simplify(dsolve( subs(F(t)=-10,{BVP||(1..4)}) )) ;
```

$$SOLFeq1 := \left\{\lambda 1(t) = \_C3, \lambda 2(t) = -\_C3\,t + t + \_C2, v(t) = t + \_C4, x(t) = \frac{1}{2}\,t^2 + \_C4\,t + \_C1\right\}$$

$$SOLFeqm1 := \{\lambda 1(t) = \_C3, \lambda 2(t) = -\_C3\,t + t + \_C2, v(t) = -10\,t + \_C4, x(t) = -5\,t^2 + \_C4\,t + \_C1\} \quad (2.5)$$

Try to start with F=1 (the right control)

```
> FIRST := simplify(dsolve( subs(F(t)=1,{BVP||(1..4),x(0) = 0, v(0)
  = 0}) )) ;
```

$$FIRST := \left\{\lambda 1(t) = \_C3, \lambda 2(t) = -\_C3\,t + t + \_C2, v(t) = t, x(t) = \frac{1}{2}\,t^2\right\} \quad (2.6)$$

use FIRST at (switching time ts) as initial condition for the second part of the solution

```
> INISECOND := subs(t=ts,FIRST) ;
```

$$INISECOND := \left\{\lambda 1(ts) = \_C3, \lambda 2(ts) = -\_C3\, ts + ts + \_C2, v(ts) = ts, x(ts) = \frac{1}{2}\, ts^2\right\} \quad \textbf{(2.7)}$$

```
> SECOND := simplify(dsolve( subs(F(t)=-10,{BVP||(1..4),op
  (INISECOND)}) )) ;
```

$$SECOND := \left\{\lambda 1(t) = \_C3, \lambda 2(t) = -\_C3\, t + t + \_C2, v(t) = -10\, t + 11\, ts, x(t) = -5\, t^2 + 11\, ts\, t - \frac{11}{2}\, ts^2\right\} \quad \textbf{(2.8)}$$

Now we must match the RIGHT boundary condition, which are v(1) =0, and lambda1(1)=0

```
> subs(t=1,subs(SECOND,v(t))) ; SOLTS := solve( %, {ts} ) ;
```

$$-10 + 11\, ts$$

$$SOLTS := \left\{ts = \frac{10}{11}\right\} \quad \textbf{(2.9)}$$

```
> subs(t=1,subs(SECOND,lambda2(t))) ; SOLC2 := solve( %, {_C2} ) ;
```

$$-\_C3 + 1 + \_C2$$

$$SOLC2 := \{\_C2 = \_C3 - 1\} \quad \textbf{(2.10)}$$

```
> FIRST := subs( SOLC2,  FIRST ) ;
```

$$FIRST := \left\{\lambda 1(t) = \_C3, \lambda 2(t) = -\_C3\, t + t + \_C3 - 1, v(t) = t, x(t) = \frac{1}{2}\, t^2\right\} \quad \textbf{(2.11)}$$

```
> SECOND := subs( SOLC2, subs( SOLTS,  SECOND )) ;
```

$$SECOND := \left\{\lambda 1(t) = \_C3, \lambda 2(t) = -\_C3\, t + t + \_C3 - 1, v(t) = -10\, t + 10, x(t) = -5\, t^2 + 10\, t - \frac{50}{11}\right\} \quad \textbf{(2.12)}$$

```
> subs(SOLTS,subs(t=ts,subs( FIRST, lambda2(t)))) ; SOLC3 := solve(
  %, {_C3} ) ;
```

$$\frac{1}{11}\, \_C3 - \frac{1}{11}$$

$$SOLC3 := \{\_C3 = 1\} \quad \textbf{(2.13)}$$

```
> FIRST  := subs( SOLC3,  FIRST ) ;
  SECOND := subs( SOLC3,  SECOND ) ;
```

$$FIRST := \left\{\lambda 1(t) = 1, \lambda 2(t) = 0, v(t) = t, x(t) = \frac{1}{2}\, t^2\right\}$$

$$SECOND := \left\{\lambda 1(t) = 1, \lambda 2(t) = 0, v(t) = -10\, t + 10, x(t) = -5\, t^2 + 10\, t - \frac{50}{11}\right\} \quad \textbf{(2.14)}$$

```
> X := piecewise( t < 10/11, subs(FIRST,x(t)), subs(SECOND,x(t)) )
  ;
```

$$X := \begin{cases} \frac{1}{2}\, t^2 & t < \frac{10}{11} \\ -5\, t^2 + 10\, t - \frac{50}{11} & otherwise \end{cases} \quad \textbf{(2.15)}$$

```
> V := piecewise( t < 10/11, subs(FIRST,v(t)), subs(SECOND,v(t)) )
  ;
```

$$V := \begin{cases} t & t < \frac{10}{11} \\ -10\,t + 10 & \textit{otherwise} \end{cases} \qquad \textbf{(2.16)}$$

```
> plot( [X,V,t=0..1] ) ;
```